

# NAG Fortran Library Routine Document

## F07ADF (SGETRF/DGETRF)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F07ADF (SGETRF/DGETRF) computes the  $LU$  factorization of a real  $m$  by  $n$  matrix.

### 2 Specification

```
SUBROUTINE F07ADF (M, N, A, LDA, IPIV, INFO)
ENTRY      sgetrf (M, N, A, LDA, IPIV, INFO)
INTEGER    M, N, LDA, IPIV(*), INFO
real      A(LDA,*)
```

The ENTRY statement enables the routine to be called by its LAPACK name.

### 3 Description

This routine forms the  $LU$  factorization of a real  $m$  by  $n$  matrix  $A$  as  $A = PLU$ , where  $P$  is a permutation matrix,  $L$  is lower triangular with unit diagonal elements (lower trapezoidal if  $m > n$ ) and  $U$  is upper triangular (upper trapezoidal if  $m < n$ ). Usually  $A$  is square ( $m = n$ ), and both  $L$  and  $U$  are triangular. The routine uses partial pivoting, with row interchanges.

### 4 References

Golub G H and van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Parameters

- 1: M – INTEGER *Input*  
*On entry:*  $m$ , the number of rows of the matrix  $A$ .  
*Constraint:*  $M \geq 0$ .
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the number of columns of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: A(LDA,\*) – **real** array *Input/Output*  
**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .  
*On entry:* the  $m$  by  $n$  matrix  $A$ .  
*On exit:*  $A$  is overwritten by the factors  $L$  and  $U$ ; the unit diagonal elements of  $L$  are not stored.
- 4: LDA – INTEGER *Input*  
*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F07ADF (SGETRF/DGETRF) is called.  
*Constraint:*  $LDA \geq \max(1, M)$ .

5: IPIV(\*) – INTEGER array Output

**Note:** the dimension of the array IPIV must be at least  $\max(1, \min(M, N))$ .

*On exit:* the pivot indices. Row  $i$  of the matrix  $A$  was interchanged with row  $IPIV(i)$  for  $i = 1, 2, \dots, \min(m, n)$ .

6: INFO – INTEGER Output

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO =  $-i$ , the  $i$ th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO =  $i$ ,  $u_{ii}$  is exactly zero. The factorization has been completed but the factor  $U$  is exactly singular, and division by zero will occur if it is subsequently used to solve a system of linear equations or to compute  $A^{-1}$ .

## 7 Accuracy

The computed factors  $L$  and  $U$  are the exact factors of a perturbed matrix  $A + E$ , where

$$|E| \leq c(\min(m, n))\epsilon P|L||U|,$$

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

## 8 Further Comments

The total number of floating-point operations is approximately  $\frac{2}{3}n^3$  if  $m = n$  (the usual case),  $\frac{1}{3}n^2(3m - n)$  if  $m > n$  and  $\frac{1}{3}m^2(3n - m)$  if  $m < n$ .

A call to this routine with  $m = n$  may be followed by calls to the routines:

F07AEF (SGETRS/DGETRS) to solve  $AX = B$  or  $A^T X = B$ ;

F07AGF (SGECON/DGECON) to estimate the condition number of  $A$ ;

F07AJF (SGETRI/DGETRI) to compute the inverse of  $A$ .

The complex analogue of this routine is F07ARF (CGETRF/ZGETRF).

## 9 Example

To compute the  $LU$  factorization of the matrix  $A$ , where

$$A = \begin{pmatrix} 1.80 & 2.88 & 2.05 & -0.89 \\ 5.25 & -2.95 & -0.95 & -3.80 \\ 1.58 & -2.69 & -2.90 & -1.04 \\ -1.11 & -0.66 & -0.59 & 0.80 \end{pmatrix}.$$

## 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      F07ADF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER         MMAX, NMAX, LDA
PARAMETER       (MMAX=8,NMAX=8,LDA=MMAX)
*      .. Local Scalars ..
INTEGER         I, IFAIL, INFO, J, M, N
*      .. Local Arrays ..
real          A(LDA,NMAX)
INTEGER         IPIV(NMAX)
*      .. External Subroutines ..
EXTERNAL        sgetrf, X04CAF
*      .. Intrinsic Functions ..
INTRINSIC       MIN
*      .. Executable Statements ..
WRITE (NOUT,*) 'F07ADF Example Program Results'
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) M, N
IF (M.LE.MMAX .AND. N.LE.NMAX) THEN
*
*      Read A from data file
*
READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
*
*      Factorize A
*
CALL sgetrf(M,N,A,LDA,IPIV,INFO)
*
*      Print details of factorization
*
WRITE (NOUT,*)
IFAIL = 0
CALL X04CAF('General',' ',M,N,A,LDA,'Details of factorization',
+          IFAIL)
*
*      Print pivot indices
*
WRITE (NOUT,*)
WRITE (NOUT,*) 'IPIV'
WRITE (NOUT,99999) (IPIV(I),I=1,MIN(M,N))
*
IF (INFO.NE.0) WRITE (NOUT,*) 'The factor U is singular'
*
END IF
STOP
*
99999 FORMAT ((3X,7I11))
END

```

## 9.2 Program Data

```

F07ADF Example Program Data
 4 4                               :Values of M and N
 1.80  2.88  2.05 -0.89
 5.25 -2.95 -0.95 -3.80
 1.58 -2.69 -2.90 -1.04
-1.11 -0.66 -0.59  0.80       :End of matrix A

```

### 9.3 Program Results

F07ADF Example Program Results

Details of factorization

	1	2	3	4
1	5.2500	-2.9500	-0.9500	-3.8000
2	0.3429	3.8914	2.3757	0.4129
3	0.3010	-0.4631	-1.5139	0.2948
4	-0.2114	-0.3299	0.0047	0.1314

IPIV

	2	2	3	4
--	---	---	---	---

---